



DECUS

PROGRAM LIBRARY

DECUS NO.	8-512a
TITLE	MODIFIED BINARY LOADER FOR PDP-8 SERIES COMPUTERS
AUTHOR	Geoffrey Chase
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	August 26, 1973
SOURCE LANGUAGE	PAL III

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

DEC 12
SANDWICH LIBRARY



RECEIVED

NOV 18 1911

1911

The following is a list of the books received from the Sandwich Library during the month of November 1911.

No.	Title	Author
1
2
3
4
5
6
7
8
9
10

BINARY LOADERS FOR PDP-8 SERIES COMPUTERS

- CORE:** 7617-7747 AND 7777, ANY FIELD. THIS LEAVES 7750-7755 FREE FOR DATA BREAK, ETC. THERE ARE ALSO 15 FREE CORE LOCATIONS (7600-7616) AT THE BEGINNING OF (OCTAL) PG. 37.
- USE:** START AT 7700 OR AT 7777. THERE IS NO BIT 0 OPTION; INSTEAD, TWO VERSIONS EXIST, EACH WITH ITS OWN TAPE, ONE "CUSTOMIZED" FOR THE LOW-SPEED (TTY) READER AND ONE FOR THE H.S.R.
- FEATURES:** THESE LOADERS FUNCTION EXACTLY LIKE THE DEC LOADER--FIELD PS.-OPS., ASCII BETWEEN RUBOUTS, ETC. ALL GET THE STANDARD TREATMENT--BUT PERFORM IN ADDITION CERTAIN CHECKS:
1. THE SECOND FRAME OF A DATA OR ADDRESS WORD IS "AND"ED WITH OCTAL 77. THIS PREVENTS SPURIOUS HIGH-ORDER BITS FROM BEING PICKED UP BY THE READER.
 2. OTHER FRAMES ARE CHECKED FOR THEIR VALUE. IF > 200, THE FRAME MUST BE OF THE FORM 3N0 (OCTAL) TO BE ACCEPTED AS FIELD PSEUDO-OP. OTHERWISE THE LOADER HALTS ON ERROR (CF. BELOW).
 3. AN ERROR IN FRAME VALUE OR IN CHECKSUM CAUSES A "JMP ." TO BE EXECUTED. THE MACHINE HANGS UP AT ONE ADDRESS WITH THE "RUN" LIGHT ON AND WILL REJECT ANY ATTEMPT TO LOAD ADDRESS, CONTINUE, OR START UNTIL IT IS MANUALLY HALTED WITH THE "HALT" (OR "STOP") SWITCH. IT IS THUS IMPOSSIBLE TO OVERLOOK A CHECKSUM ERROR.
 4. THE TAPE FORMATS AND LOADING PROCEDURES ARE DESCRIBED SEPARATELY BEFORE EACH LISTING. BOTH TAPES USE A FEW DIRTY TRICKS TO ENABLE THE RIM LOADER TO LOAD A BIN-FORMAT TAPE. THIS SAVES ABOUT HALF THE LOADING TIME (OF SOME VALUE WITH LOW-SPEED INPUT). WHEN TRAILER IS REACHED, THE RIM LOADER IS RESTORED, THE TAPE READER STOPS, AND THE PROCESSOR HALTS AT 7677.

SEE ACCOMPANYING WRITE-UPS FOR:

1. LOADING THESE TAPES WITH A BIN LOADER (DESCRIBED IN PREFACE TO THE LOW-SPEED VERSION)
2. LOADING HIGH-SPEED RIM AND BIN WITH A "HELP" BOOTSTRAP (DESCRIBED IN PREFACE TO THE HIGH-SPEED VERSION)

PDP-8 BINARY LOADER, LOW-SPEED READER VERSION

THE TAPE SUPPLIED IS IN RIM/BIN FORMAT; IT IS ACTUALLY
A STANDARD BINARY OUTPUT TAPE FROM PAL-III.

NORMAL LOADING PROCEDURE:

1. CHECK TO SEE THAT THE RIM LOADER IS IN CORE.
2. LOAD THE BINARY LEADER (CODE 200, ONE EDGE PUNCHED)
INTO THE TTY PAPER TAPE READER. TURN TTY TO "LINE",
READER CONTROL TO "START".
3. LOAD ADDRESS = 7756 AND START THE PROCESSOR. THE
TAPE WILL HALT ITSELF WHEN TRAILER IS REACHED; THE
PROCESSOR WILL HALT AT ADDRESS 7677.

[WITH THE PDP-8/E AND LATER MODELS, THIS WILL LEAVE 7700
IN THE MEMORY ADDRESS LIGHTS. SINCE THE BINARY LOADER
CAN BE STARTED AT EITHER 7777 OR 7700, YOU ARE ALL SET
TO GO.]

LOADING WITH A BINARY LOADER:

IN THE UNLIKELY EVENT THAT YOU WISH TO DO SO, THE TAPE
SUPPLIED CAN BE LOADED WITH A BINARY LOADER. THE CHECKSUM
(IF THE READER MAKES NO ERRORS) WILL BE 0, AS USUAL.

THIS CAN ONLY BE DONE, OF COURSE, IF THE DATA FIELD (INTO
WHICH THIS LOADER IS DEPOSITED) DIFFERS FROM THE INSTRUCTION
FIELD (WHERE THE EXISTING BINARY LOADER RESIDES). OTHER-
WISE THE BINARY LOADER IN USE IS OVERLAID BY THE BINARY LOADER
WHICH IT IS LOADING, WITH UNPREDICTABLE--USUALLY RUINOUS--
RESULTS.

/ BIN LOADER 8/73
/PDP-8 SERIES COMPUTERS, LOW-SPD. READER ONLY

/THE MATERIAL ON THIS PAGE IS USED TO BOOT IN THE LOADER:

		*7752		
7752	3776	LOOPRM, DCA I ADRS		
		*7753		
7753	2376	ISZ ADRS		
		*7754		
7754	7410	SKP		/UNTIL ADRS OVERFLOWS (7777+1)
		*7755		
7755	5214	JMP 7614		/DONE LOADING, RESTORE RIM
		*7613		
7613	3776	K3776, 3776		/USED TO RESTORE RIM LOADER
		*7614		
7614	1213	TAD K3776		
		*7615		
7615	3373	DCA 7773		
		*7616		/LOCNS. 7613-7616 & 7752-7755
7616	5277	JMP 7677		/ ARE USED ONLY TO BOOT THIS
				/ LOADER INTO CORE. AFTERWARDS,
				/ THEY ARE FREE FOR OTHER USES
				/ (E.G., DATA BREAK) & WILL NOT
				/ BE OVERLAID.
		*7773		
7773	5352	JMP LOOPRM		/MAKES RIM INTO A DIRTY BIN LDR.

FIRST= 7617 /LOWEST LOCATION USED BY THIS BIN. LOADER
 SECOND=7620
 MODE= 7621

*7622

7622	0000	FRAME1, 0		/SUBR. FOR ODD FRAMES
7623	3221		DCA MODE	
7624	4263		JMS READ	
7625	1277		TAD M376	
7626	7750		SPA SNA CLA	
7627	5233		JMP NONRBT	
7630	2221		ISZ MODE	/RUBOUT: COMPLEMENT "MODE"
7631	7040		CMA	
7632	5223		JMP FRAME1+1	
7633	1221	NONRBT,	TAD MODE	/DATA MODE=0, ASCII MODE =-1
7634	7640		SZA CLA	
7635	5224		JMP FRAME1+2	
7636	1347		TAD SAV	
7637	1244		TAD M200	
7640	7510		SPA	
7641	2222		ISZ FRAME1	/FRAME < 200? RET. TO CALL+2
7642	7540		SMA SZA	
7643	5246		JMP FLD	/CODE > 200
7644	7600	M200,	7600	
7645	5622		JMP I FRAME1	
7646	1262	FLD,	TAD M100	
7647	7500		SMA	
7650	0333		AND K7	
7651	7640		SZA CLA	
7652	5252		JMP .	/EITHER 200<FRAME<300, OR /FRAME DOESN'T END IN OCTAL 0
7653	1347		TAD SAV	
7654	0260		AND K70	
7655	1344		TAD KCDF	
7656	3346		DCA DATAF	
7657	5224		JMP FRAME1+2	/GET A NEW 1ST FRAME
7660	0070	K70,	70	
7661	0077	K77,	77	
7662	7700	M100,	-100	
7663	0000	READ,	0	
7664	6031		KSF	
7665	5264		JMP .-1	
7666	6036		KRB	
7667	3347		DCA SAV	
7670	1347		TAD SAV	
7671	5663		JMP I READ	

7672	4335	TRAILR,	JMS PACK	/REACHED END OF TAPE
7673	7041		CIA	
7674	1345		TAD CKSUM	
7675	7440		SZA	
7676	5276		JMP .	/BAD CKSUM
7677	7402	M376,	HLT	/GOOD CKSUM
7700	6032	BEG IN,	KCC	
7701	6214		RDF	
7702	1344		TAD KCDF	
7703	3346		DCA DATAF	
7704	4222		JMS FRAME1	
7705	5304		JMP .-1	/LEADER
7706	3345	LOOP1,	DCA CKSUM	
7707	1346		TAD DATAF	
7710	3330		DCA DEPOST	/D.F. FOR CURRENT FRAME
7711	1347		TAD SAV	
7712	3217		DCA FIRST	/FIRST FRAME OF WD.
7713	4263		JMS READ	
7714	0261		AND K77	
7715	3220		DCA SECOND	/SECOND FRAME
7716	4222		JMS FRAME1	
7717	5272		JMP TRAILR	/IF CODE 200
7720	4335		JMS PACK	/NOT 200, COMBINE FRAMES
7721	7420		SNL	
7722	5330		JMP DEPOST	/DATA WORD
7723	3376		DCA ADRS	/ADRS WORD
7724	1217	LOOP2,	TAD FIRST	
7725	1220		TAD SECOND	
7726	1345		TAD CKSUM	/UPDATE CKSUM
7727	5306		JMP LOOP1	
7730	6201	DEPOST,	CDF	
7731	3776		DCA I ADRS	
7732	2376		ISZ ADRS	
7733	0007	K7,	7	/PROTECT ISZ
7734	5324		JMP LOOP2	
7735	0000	PACK,	0	
7736	1217		TAD FIRST	
7737	7106		CLL RTL	
7740	7006		RTL	
7741	7006		RTL	
7742	1220		TAD SECOND	
7743	5735		JMP I PACK	/LINK=1 IF AN ADRS. WORD
7744	6201	KCDF,	CDF	
		CKSUM=	7745	
		DATAF=	7746	
		SAV=	7747	/7750-7755 LEFT FREE

/ RIM (LOW-SPEED) IN USUAL PLACE

*7777

7777 5300 JMP BEGIN

ADRS=7776

/ADRS PTR. FOR BOTH LOADERS

/SUPPLIED AS A TAPE IN RIM/BIN FORMAT.
/TAPE MODIFIES RIM LOADER, LOADS ITSELF,
/RESTORES RIM LOADER, HALTS PAPER TAPE
/READER, AND HALTS PROCESSOR AT *7677

ADRS	7776
BEGIN	7700
CKSUM	7745
DATAF	7746
DEPOST	7730
FIRST	7617
FLD	7646
FRAME1	7622
KCDF	7744
K3776	7613
K7	7733
K70	7660
K77	7661
LOOPRM	7752
LOOP1	7706
LOOP2	7724
MODE	7621
M100	7662
M200	7644
M376	7677
NONRBT	7633
PACK	7735
READ	7663
SAV	7747
SECOND	7620
TRAILR	7672

PDP-8 BINARY LOADER, HIGH-SPEED READER VERSION

THE TAPE SUPPLIED CONSISTS OF:

- A. BLANK LEADER
- B. FRANK PALMISANO'S "HELP" LOADER (ODD-LOOKING !)
- C. ABOUT 1 INCH OF CODE 200 (BINARY LEADER)
- D. THE BINARY LOADER PROGRAM, IN RIM/BIN FORMAT

THIS TAPE CAN BE LOADED BY THE STANDARD RIM LOADER,
BY A BIN LOADER [IF IT IS LOADED OFF-FIELD], OR BY A
SPECIAL "HELP" BOOTSTRAP.

TO LOAD WITH RIM OR A BIN LOADER, FIND SECTION C. (CODE 200'S)
AND PLACE IT IN THE HIGH-SPEED READER. PROCEED AS USUAL.

IF NO LOADER IS IN CORE, TRY THE FOLLOWING:

- 1. LOAD SECTION A. (BLANKS) INTO THE HIGH-SPEED READER.
- 2. TOGGLE IN THE BOOTSTRAP SHOWN BELOW.
- 3. START PROCESSOR AT 0000 IN FIELD 0.

[THIS TECHNIQUE WILL LOAD INTO FIELD 0 ONLY.
INTERRUPT IS USED; AVOID STRIKING TTY KEYS., ETC.]

BOOTSTRAP:

LOCATION:	TOGGLE IN:
0000	6014
1	6016
2	6001
3	7530
4	3410
5	7106
6	5006
7	0000
10	0000

/ BIN LOADER 8/73
/PDP-8 SERIES COMPUTERS, HIGH-SPD. READER ONLY

/THE MATERIAL ON THIS PAGE IS USED TO BOOT IN THE LOADER:

		*7751		
7751	3776	LOOPRM, DCA I ADRS		
		*7752		
7752	2376	ISZ ADRS		
		*7753		
7753	7410	SKP		/UNTIL ADRS OVERFLOWS (7777+1)
		*7754		
7754	5214	JMP 7614		/DONE LOADING, RESTORE RIM
		*7755		
7755	7410	SKP		/HIGH-SPD. RIM RETURNS TO 7757
		*7613		
7613	3776	K3776, 3776		/USED TO RESTORE RIM LOADER
		*7614		
7614	1213	TAD K3776		
		*7615		
7615	3373	DCA 7773		
		*7616		
7616	5277	JMP 7677		/LOCNS. 7613-7616 & 7751-7755 / ARE USED ONLY TO BOOT THIS / LOADER INTO CORE. AFTERWARDS, / THEY ARE FREE FOR OTHER USES / (E.G., DATA BREAK) & WILL NOT / BE OVERLAID.
		*7773		
7773	5351	JMP LOOPRM		/MAKES RIM INTO A DIRTY BIN LDR.

FIRST= 7617 /LOWEST LOCATION USED BY THIS BIN. LOADER
 SECOND=7620
 MODE= 7621

*7622

7622	0000	FRAME1, 0		/SUBR. FOR ODD FRAMES
7623	3221		DCA MODE	
7624	4263		JMS READ	
7625	1277		TAD M376	
7626	7750		SPA SNA CLA	
7627	5233		JMP NONRBT	
7630	2221		ISZ MODE	/RUBOUT: COMPLEMENT "MODE"
7631	7040		CMA	
7632	5223		JMP FRAME1+1	
7633	1221	NONRBT,	TAD MODE	/DATA MODE=0, ASCII MODE =-1
7634	7640		SZA CLA	
7635	5224		JMP FRAME1+2	
7636	1347		TAD SAV	
7637	1244		TAD M200	
7640	7510		SPA	
7641	2222		ISZ FRAME1	/FRAME < 200? RET. TO CALL+2
7642	7540		SMA SZA	
7643	5246		JMP FLD	/CODE > 200
7644	7600	M200,	7600	
7645	5622		JMP I FRAME1	
7646	1262	FLD,	TAD M100	
7647	7500		SMA	
7650	0333		AND K7	
7651	7640		SZA CLA	
7652	5252		JMP .	/EITHER 200<FRAME<300, OR /FRAME DOESN'T END IN OCTAL 0
7653	1347		TAD SAV	
7654	0260		AND K70	
7655	1346		TAD KCDF	
7656	3345		DCA DATAF	
7657	5224		JMP FRAME1+2	/GET A NEW 1ST FRAME
7660	0070	K70,	70	
7661	0077	K77,	77	
7662	7700	M100,	-100	
7663	0000	READ,	0	
7664	6011		RSF	
7665	5264		JMP .-1	
7666	6016		RRB RFC	
7667	3347		DCA SAV	
7670	1347		TAD SAV	
7671	5663		JMP I READ	

7672	4335	TRAILR,	JMS PACK	/REACHED END OF TAPE
7673	7041		CIA	
7674	1344		TAD CKSUM	
7675	7440		SZA	
7676	5276		JMP .	/BAD CKSUM
7677	7402	M376,	HLT	/GOOD CKSUM
7700	6014	BEGIN,	RFC	
7701	6214		RDF	
7702	1346		TAD KCDF	
7703	3345		DCA DATAF	
7704	4222		JMS FRAME1	
7705	5304		JMP .-1	/LEADER
7706	3344	LOOP1,	DCA CKSUM	
7707	1345		TAD DATAF	
7710	3330		DCA DEPOST	/D.F. FOR CURRENT FRAME
7711	1347		TAD SAV	
7712	3217		DCA FIRST	/FIRST FRAME OF WD.
7713	4263		JMS READ	
7714	0261		AND K77	
7715	3220		DCA SECOND	/SECOND FRAME
7716	4222		JMS FRAME1	
7717	5272		JMP TRAILR	/IF CODE 200
7720	4335		JMS PACK	/NOT 200, COMBINE FRAMES
7721	7420		SNL	
7722	5330		JMP DEPOST	/DATA WORD
7723	3376		DCA ADRS	/ADRS WORD
7724	1217	LOOP2,	TAD FIRST	
7725	1220		TAD SECOND	
7726	1344		TAD CKSUM	/UPDATE CKSUM
7727	5306		JMP LOOP1	
7730	6201	DEPOST,	CDF	
7731	3776		DCA I ADRS	
7732	2376		ISZ ADRS	
7733	0007	K7,	7	/PROTECT ISZ
7734	5324		JMP LOOP2	
7735	0000	PACK,	0	
7736	1217		TAD FIRST	
7737	7106		CLL RTL	
7740	7006		RTL	
7741	7006		RTL	
7742	1220		TAD SECOND	
7743	5735		JMP I PACK	/LINK=1 IF AN ADRS. WORD
7744	0000	CKSUM,	0	
7745	0000	DATAF,	0	
7746	6201	KCDF,	CDF	
7747	0000	SAV,	0	
				/ RIM (HIGH-SPEED) IN USUAL PLACE
				*7777
7777	5300		JMP BEGIN	
				ADRS=7776
				/ADRS PTR. FOR BOTH LOADERS

ADRS	7776
BEGIN	7700
CKSUM	7744
DATAF	7745
DEPOST	7730
FIRST	7617
FLD	7646
FRAME1	7622
KCDF	7746
K3776	7613
K7	7733
K70	7660
K77	7661
LOOPRM	7751
LOOP1	7706
LOOP2	7724
MODE	7621
M100	7662
M200	7644
M376	7677
NONRBT	7633
PACK	7735
READ	7663
SAV	7747
SECOND	7620
TRAILR	7672

